
	1. Введение

Моделирование, как метод исследования сложных объектов, в условиях использования цифровых вычислительных машин имеет ряд особенностей:

1. При моделировании временных объектов шкала времени, в отличие от реального объекта, имеет дискретный, квантованный характер. 

2. В качестве источников сигнала или информации используют программы‑генераторы случайных последовательностей с заданным законом распределения.

3. Результаты исследования модели также носят дискретный характер и могут быть выражены через статистические параметры процессов модели (средние, дисперсии) или в виде таблиц, содержащих временные ряды состояний отдельных объектов модели.

В качестве кванта времени при моделировании быстротекущих процессов (химические реакторы, управление транспортом) используют мили- и микросекунды; по обслуживанию населения — минуты и секунды, макроэкономических систем — дни и годы.

В качестве источника сигнала чаще всего используют генераторы сигналов с пуассоновским законом распределения. Вероятность появления k событий за время t в этом случае будет иметь вид:

Pk(t) = 

Где ( — параметр распределения. При это среднее значение и дисперсия равны (.

Длительность времени между появлениями двух последовательных событий простейшего потока (случайная величина T) распределена по экспоненциальному закону. В этом случае интеграл вероятности появления одного события через время t будет иметь вид:

F(t) = 1 – e–(t.
Последнее соотношение является основой для создания генераторов последовательностей пуассоновского потока.

	2. Постановка задачи

В качестве основы для постановки задачи используем задание на курсовой проект.

Задание

Автомобили с товаром подъезжают к складу с интенсивностью 12 автомобилей в час в соответствии с пуассоновским законом распределения обслуживания является экспоненциальным, но среднее время обслуживания зависит от числа автомобилей, находящихся в очереди. Эта зависимость представлена в таблице:

                  Длина                   Среднее время

                очереди                  обслуживания

0 5,5

                   1 и 2                           5,0

                 3,4 или 5                      4,5

                   6 и более                    4,0

Построить модель, которая давала бы информацию об очереди автомобилей и времени простоя склада. Выполнение моделирования должно охватить 5 отдельных пятичасовых дней.

Постановка задачи – точная формулировка подлежащей решению задачи.

Из задания, следует:

1. В качестве исходных данных используют:

· количество часов в рабочем дне,

· среднее количество приезда автомобилей на склад за час,

· количество рабочих дней в неделе (время моделирования),

· среднее время обслуживания автомобиля складом,

· длительность такта модели.

2. Модель имеет генератор последовательностей пуассоновского потока. С одним параметром распределения формируется очередь автомобилей на склад (Г1).  С другим параметром распределения определяется время обслуживания складом автомобиля (Г2). Причем время обслуживания таблично связано с очередью автомобилей на склад.

	3. Модель запускается для каждого дня со стартового времени t = 0 .

4. Генератором Г1 определяется время приезда очередного автомобиля на склад.

5. При увеличении текущего времени на 1 сек выполняют проверку наступления событий в модели. Такими событиями являются:

· приезд автомобиля на склад,

· если автомобиль приехал на склад, то выполняется постановка приехавшего автомобиля в общую очередь и определяется время приезда следующего автомобиля;

· проверка занятости пункта приема автомобилей на складе,

· если пункт приема автомобилей на складе занят, то уменьшают его время на обслуживание автомобиля на 1 сек;

· если пункт приема автомобилей незанят и очередь пуста, то увеличивают время простоя склада по текущему дню;

· если пункт приема автомобилей на складе незанят и очередь непустая, то  генератором Г2  определяют время с учётом длины очереди, которое будет потрачено на обслуживание автомобиля, записывают это время за пунктом приема склада и уменьшают значение очереди автомобилей на склад на единицу;

· начало нового интервала времени для формирования данных динамики максимального значения количества автомобилей в очереди по контрольным точкам,

· запись начального значения максимума количества автомобилей в очереди на интервале,

· определение максимального значения количества автомобилей в очереди на интервале контрольной точки динамики,

· определение максимального значения количества автомобилей в очереди за день,

· окончание рабочего дня,

· окончание процесса моделирования по заданному количеству рабочих дней,

Учет простоя склада ведется по каждому отработанному дню.

	6. В качестве результата выдается (см.таблицу):

· время простоя склада по дням

· динамика очереди автомобилей по максимуму в течение всех дней,

· максимальное количество автомобилей в очереди за каждый день,

· количество автомобилей, приезжавших на склад по каждому дню.

                                                                                                  Таблица

Результаты испытания модели обслуживания автомобилей на складе

Математическая модель – система математических зависимостей, описывающих структуру или функционирование объекта.

Для составления математической модели присвоим имена всем данным, которые будут участвовать в решении задачи. Сделаем это с учётом макета таблицы результатов исследования модели (см.таблицу), что позволит в любой момент легко установить соответствие между графами таблицы и именами находящихся в этих графах данных. Список имен данных имеет вид:

   WorkDayHour —   количество часов в рабочем дне,
   AvtoDayMidl —  среднее количество приезда автомобилей в час,

   Wday  —                количество рабочих дней в неделе (время моделирования),

   SkladCnt —            пунктов приема автомобилей на складе,

   SkladMidl —          время обслуживания автомобиля на складе,

   PntNum —             количество контрольных точек за день,

	   WorkDaySec —     количество секунд в рабочем дне,

   AvtoMidl —        среднее время приезда одного автомобиля на склад,

   LambdaAvto —    среднее количество приездов автомобилей за одну сек,

   LambdaSklad —     кол-во обслуживаемых автомобилей складом за 1 сек,

   PntNumSec —       количество сек для расчёта  контрольной точки,

   Id —                       счётчик дней,

   Is —                       счётчик секунд в одном дне,

   I,j —                       переменные циклов,

   TimeAvto —        время, через которое появится на складе очередной автомобиль,

   CntAvto —          текущее количество автомобилей в очереди,

   Npnt —                  номер контрольной точки динамики очереди автомобилей,

   SkladN[1..SkladCnt,1..Wday+1] —  массив учёта работы пунктов обслуживания автомобилей на складе (В алгоритме и программе закладываем возможность её модернизации для построения моделей склада с несколькими пунктами приёма. В задании предполагается один пункт приёма автомобилей на складе),

   SkladN[1..SkladCnt,1] —  элемент массива работы пункта приёма автомобилей на складе, содержащий текущую загрузку (оставшееся время) обслуживания автомобиля;

   SkladN[1..SkladCnt,1+i] —  элемент массива  работы пунктов приёма автомобилей на складе, содержащий время простоя пункта приёма автомобилей на складе за i-тый день;
   Avto[1..Wday] —  массив количества автомобилей за день,

   AvtoMax[1..Wday] —массив максимального количества автомобилей в очереди за день,

   AvtoMaxPnt [1..Wday,1..PntNum] —    массив максимального количества автомобилей в очереди контрольной точки динамики по текущему дню,

  Lambda —             параметр распределения,

   M —                       идентификатор файла,

   Q —                       имя файла,

   Rday[1..Wday] —  массив  среднего  значения простоя склада за день.

   Rand (Lambda) — функция вычисления случайной величины распределенной по экспоненциальному закону.


	
	КП.2203.42.23.99.01 ПЗ
	Лист

	
	
	8

	Исходные данные

WorkDayHour, AvtoDayMidl, Wday, SkladCnt, SkladMidl, PntNum — задаются в виде констант модели.

Расчётные данные в виде констант

WorkDaySec=WorkDayHour*60*60;

AvtoMidl=(60.0*60.0)/AvtoDayMidl;

LambdaAvto=1.0/AvtoMidl;  

LambdaSklad=1.0/SkladMidl;

PntNumSec=Целая часть от (WorkDaySec/PntNum);

Расчётные данные случайной величины

Rand(Lambda)= – ln(1.0-Random)/Lambda,

      TimeAvto = – ln(1.0-Random)/LambdaAvto, 

Где Random — случайная величина, имеющая равномерный закон распределения. Соотношение реализует пуассоновский поток в соответствии с теорией, изложенной в разделе 1.
Расчётные данные итоговые

SkladNij[m,1] = Randj(Г2)– 

где j — номер автомобиля, i — текущее время обслуживания автомобиля в сек на m–том пункте обслуживания склада.

SkladN[m,1+i] =  

где Rj, Rk, … — время j–го, k–го и … простоев за i–тый день m–го пункта приёма автомобилей на складе.

Avto[i] = 

где Kl — количество точек потока автомобилей за i-тый день, полученных с генератора Г1.
AvtoMax[i] = maxAvtoCntj ({AvtoCntj|j = 

где j — время i–го дня.

	AvtoMaxPnt [i,k] = maxAvtoCntjk ({AvtoCntjk|j = 
[image: image7.wmf]PntNumSec

где k — номер контрольной точки, 

       i — номер дня,

       j — время интервала контрольной точки

SkladN[m,1+i]
Rday[i] =  

Где m — номер пункта приема автомобилей на складе, 

       i — номер дня.
Расчёт номера контрольной точки динамики

Npnt = целая часть от (Is/PntNumSec+1);

Типы данных, участвующих в решении задачи:

WorkDayHour, AvtoDayMidl,Wday, SkladCnt, SkladMidl — (SkladMidl0, SkladMidl2, SkladMidl5, SkladMidl6); PntNum,WorkDaySec, AvtoMidl, LambdaAvto, LambdaSklad — (LambdaSklad0, LambdaSklad2, LambdaSklad5, LambdaSklad6), PntNumSec — числовые переменные вещественного типа;

Id, Is, i, j, TimeAvto, CntAvto, Npnt — числовые переменные целого типа;                

SkladN[1..SkladCnt,1..Wday+1], Avto[1..Wday], AvtoMax[1..Wday, 1..PntNum], AvtoMaxPnt [1..Wday,1..PntNum] — двумерные числовые массивы целого типа
SkladN[1..SkladCnt,1], Rday[1..Wday] — одномерные числовые массивы целого типа;

Rand (Lambda) — функция вещественного типа.

Пользуясь принятыми обозначениями, составим блок схему алгоритма решения задачи.

	3. Схема программы, описание схемы программы

Алгоритм – точное предписание, определяющее вычислительный процес, ведущий от варьируемых начальных данных к искомому результату.

Алгоритм решения задачи может быть представлен разными способами, в том числе и в виде блок–схемы. Блоки, используемые для составления блок-схемы, должны соответствовать ГОСТ 19.701–90. Форма блока описывает совершаемое действие, а содержимое блока – те объекты (в нашем случае имена переменных), над которыми данное действие совершается.

Схема алгоритма задачи

В приведенной блок-схеме блоки имеют следующее содержание.

Символ 1 — начало алгоритма задачи.

Символ 2 — подключение модуля управления экраном.

Символ3 — задание и расчёт констант. Список констант на с.9.

Символ 4 — описание переменных. Список переменных приведен на с.7–8.

Символ 5 — выполнение просчёта модели.

Символ 6 — выдача результатов расчёта модели.

Символ 7 — конец алгоритма задачи.

	4. Программа

Программа – это последовательность предложений языка программирования, описывающих алгоритм решения задачи.

Текст программы структурирован. Операторы написаны с разным отступом от номера строки. Такая структуризация делает программу более удобной для чтения и анализа.
4.1. Обоснование выбора языка программирования

В качестве языка программирования для разработанного алгоритма выбран Turbo Pascal 5.0. Его возможности позволяют выполнить все требования к реализации поставленной задачи. Он обладает интегрированной оболочкой подготовки, редактирования, компиляции и отладки программ. Кроме того, эта оболочка занимает немного места и может быть реализована практически на всем семействе IBM совместимых компьютеров с объемом оперативной памяти не менее 1Mb и объёмом свободно дисковой памяти не менее 3 Mб. Язык Turbo Pascal 5.0 широко используют как учебный язык структурного программирования. Всё это послужило обоснованием в сторону выбора для решения предложенной задачи языка программирования Turbo Pascal 5.0.

4.2. Распечатка программы

{ copyright (с) Трушина М.А.,1999, МКУ и НТ, Москва,

                Turbo Pascal 5.0

            Курсовой проект по компьютерному моделированию

Тема:

     Построение модели обслуживания автомобилей на складе

Описание модели:

     Автомобили с товаром подъезжают к складу с интенсивностью

     12 автомобилей в час в соответствии с пуассоновским законом

     распределения.

     Обслуживание является экспоненциальным, но среднее время

	обслуживания зависит от числа автомобилей, находящихся в

     очереди.

     Эта зависимость представлена в таблице:

г============T===============¬

    ¦   Длина    ¦  Среднее время¦

    ¦  очереди   ¦  обслуживания ¦

    ¦============+===============¦

    ¦   0        ¦     5,5       ¦

    ¦============+===============¦

    ¦ 1 или 2    ¦     5,0       ¦

    ¦============+===============¦

    ¦ 3,4 или 5  ¦     4,5       ¦

    ¦============+===============¦

    ¦ 6 и более  ¦     4,0       ¦

    L============¦===============-

Задание:

     Построить модель, которая давала бы информацию об очереди

     автомобилей и времени простоя склада.

     Выполнение моделирования должно охватывать 5 отдельных

     пятичасовых дней.

}

Program Sklad;

Uses Crt;

Const

   WorkDayHour=5;                {Количество часов в рабочем дне}

   AvtoDayMidl=12;              {Среднее количество автомобилей в час}

   Wday=5;                       {Количество рабочих дней в неделе}

   SkladCnt=1;                   {Количество пунктов обслуживания на складе}

	SkladMidl0=5.5*60.0;           {Среднее время обслуживания автомобиля

                                  складом при отсутствии очереди, сек}

   SkladMidl2=5.0*60.0;           {Среднее время обслуживания автомобиля

                                  складом при количестве автомобилей

                                  в очереди 1 или 2, сек}

   SkladMidl5=4.5*60.0;           {Среднее время обслуживания автомобиля

                                  складом при количестве автомобилей

                                  в очереди 3, 4 и 5, сек}

   SkladMidl6=4.0*60.0;           {Среднее время обслуживания автомобиля

                                  складом при количестве автомобилей

                                  в очереди 6 и более, сек}

   PntNum=20;                    {Количество контрольных точек за день}

   WorkDaySec=WorkDayHour*60*60; {Количество секунд в рабочем дне}

   AvtoMidl=(60.0*60.0)/AvtoDayMidl;

                                 {Среднее время приезда одного

                                  автомобиля на склад, сек}

LambdaAvto=1.0/AvtoMidl;      {Среднее кол-во приезда автомобилей на склад

                                  за одну сек}

   LambdaSklad0=1.0/SkladMidl0;   {Среднее кол-во обслуживаемых автомобилей

                                  при отсутствии автомобиле в очереди

                                  за одну сек}

LambdaSklad2=1.0/SkladMidl2;    {Среднее кол-во обслуживаемых автомобилей

                                  складом при количестве автомобилей

                                  в очереди 1 или 2

                                  за одну сек}

	LambdaSklad5=1.0/SkladMidl5;    {Среднее кол-во обслуживаемых автомобилей

                                  складом при количестве автомобилей

                                  в очереди 3, 4 и 5

                                  за одну сек}

   LambdaSklad6=1.0/SkladMidl6;    {Среднее кол-во обслуживаемых автомобилей

                                  складом при количестве автомобилей

                                  в очереди 6 и более

                                  за одну сек}

   PntNumSec=Trunc(WorkDaySec/PntNum);

                                  {Количество сек для расчёта

                                   контрольной точки}

var

   Id,    {Счётчик дней}

   Is,    {Счётчик секунд в одном дне}

   i,j,   {Переменные циклов}

   TimeAvto, {Время, через которое приедет на склад очередной автомобиль}

   CntAvto,  {Текущее количество автомобилей в очереди }

   Npnt        {Номер контрольной точки динамики очереди автомобилей}

      :longint;

            {Массив учёта работы склада}

   SkladN :array [1..SkladCnt,1..1+Wday] of longint;

   Avto,       {Количество автомобилей за день}

   AvtoMax     {Максимальное количество автомобилей в очереди за день}

      :array [1..Wday] of longint;

	AvtoMaxPnt  {Максимальное количество автомобильов в очереди

                  контрольной точки динамики по текущему дню}

      :array [1..Wday,1..PntNum] of longint;

   LambdaSklad

   :real;    {Среднее кол-во обслуживаемых автомобилей

              складом за одну сек}

{--------------------------------------------------------------------}

{Функция-генератор  отметок с экспоненциальным распределением}

Function Rand(Lambda:real):single;  {Lambda - параметр распределения}

   begin

      Rand:=-ln(1.0-Random)/Lambda;

   end;

{--------------------------------------------------------------------}

Procedure model;

   Begin

       Randomize;  {Инициализация датчика случайных чисел}

{ Очистка данных по каждому из пунктов обработки автомобилей на складе}

       for j:=1 to 1+Wday do for i:=1 to SkladCnt do SkladN[i,j]:=0;

{ Заголовок цикла перебора дней}

       for Id:=1 to Wday do

          begin

             Avto[Id]:=0; {Начальное значение автомобилей за день}

             CntAvto:=0;           {Нач. знач. автомобилей в очереди}

             AvtoMax[Id]:=0; {Нач.знач.максимума автомобилей.в очереди}

for i:=1 to SkladCnt do SkladN[i,1]:=0;{ Нач.знач. загрузки
                 пунктов обработки автомобилей на складе}

{Нач.знач.интервала прихода очередного автомобия на склад}

             TimeAvto:=Round(Rand(LambdaAvto));

	{ Заголовок цикла перебора секунд за один день}

             for Is:=0 to WorkDaySec-1 do

              begin

{Проверка: Приехал ли очередной автомобиль на склад?}

                if TimeAvto<=Is then

                   begin

                     CntAvto:=CntAvto+1;  {Увеличение очереди
                                         автомобилей на один  автомобиль}

{Корректировка значения максимального количества автомобилей в очереди}

                     If CntAvto>AvtoMax[Id]

                        then AvtoMax[Id]:=CntAvto;

{Подсчёт общего количества автомобилей, прибывших на склад за день}

                     Avto[Id]:=Avto[Id]+1;

{Расчёт времени прихода следующего автомобиля в склад}

                     TimeAvto:=Round(Is+Rand(LambdaAvto))

                   end;

{Расчет номера контрольной точки динамики очереди}

                Npnt:=trunc(Is/PntNumSec+1);

{Корректировка ошибки, связанной с округлением,  последнего номера

контрольной точки}

                if Npnt>PntNum then Npnt:=PntNum;

{Установка начального максимального значения количества автомобилей в

 очереди в интервале контрольной точки динамики очереди}

                if Is=PntNumSec*(Npnt-1) then

                   AvtoMaxPnt[Id,Npnt]:=CntAvto;

{Корректировка значения максимального количества автомобилей в очереди

 на интервале контрольной точки}

If CntAvto>AvtoMaxPnt[Id,Npnt]

                   then AvtoMaxPnt[Id,Npnt]:=CntAvto;

	{Заголовок цикла выполнения действий по массиву складов}

                for i:=1 to SkladCnt do

                  begin

                   if CntAvto>0 then    {Проверка наличия автомобилей

                                           в очереди}

                      begin

{Проверка условия отстутствия автомобилей у пункта обслкживания на  складе}

                        if SkladN[i,1]=0 then

{Установка времени обслуживания автомобиля в зависимости

от величины очереди}

                           begin

                              case CntAvto of

                                 0:LambdaSklad:=LambdaSklad0;

                                 1:LambdaSklad:=LambdaSklad2;

                                 2:LambdaSklad:=LambdaSklad2;

                                 3:LambdaSklad:=LambdaSklad5;

                                 4:LambdaSklad:=LambdaSklad5;

                                 5:LambdaSklad:=LambdaSklad5;

                              else

                                 LambdaSklad:=LambdaSklad6;

                              end;

{Постановка автомобилей на обслуживание складом}

                              SkladN[i,1]:=Round(Rand(LambdaSklad));

{Уменьшение очереди автомобилей на один автомобиль}

                              CntAvto:=CntAvto-1

                           end

                        else

{Отсчет времени обслуживания автомобиля складом}

                           SkladN[i,1]:=SkladN[i,1]-1

                      end

	else

                      begin

{Проверка условия отстутствия автомобильов у склада}

                        if SkladN[i,1]=0 then

{Увеличение счётчика простоя склада на одну секунду}

                           SkladN[i,Id+1]:=SkladN[i,Id+1]+1

                        else

{Отсчет времени обслуживания автомобильа складом}

                           SkladN[i,1]:=SkladN[i,1]-1

                      end

                  end

              end

          end

   end;

Procedure wywod;

   var

     m:text;                    {Идентификатор файла}

     Q:string;                  {Имя файла}

{Массив  среднего  значения простоя складов за день}

     Rday :  array [1..Wday] of  longint;

BEGIN

{Открытие файла для записи результатов работы программы}

    clrScr;

    gotoxy(10,4);

    writeln('                   ЗАПИСЬ НА ДИСК');

    gotoxy(10,10);

    writeln('----------------------------------------------------------¬ ');

gotoxy(10,11);

    writeln('¦ВВЕДИТЕ ИМЯ ФАЙЛА:                                       ¦ ');

    gotoxy(10,12);

    writeln('L---------------------------------------------------------- ');

gotoxy(30,11);

     readln(Q);

ASSIGN(m,Q);

	  REwrite(m);

writeln(m,'            Курсовой проект по компьютерному моделированию      ');

writeln(m,'Тема:                                                           ');

writeln(m,'     Построение модели обслуживания автомобилей на складе       ');

writeln(m,'Описание модели:                                                ');

writeln(m,'     Автомобили с товаром подъезжают к складу с интенсивностью  ');

writeln(m,'     12 автомобилей в час в соответствии с пуассоновским законом');

writeln(m,'     распределения.                                             ');

writeln(m,'     Обслуживание является экспоненциальным, но среднее время   ');

writeln(m,'     обслуживания зависит от числа автомобилей, находящихся в   ');

writeln(m,'     очереди.                                                   ');

writeln(m,'     Эта зависимость представлена в таблице:                    ');

writeln(m,'    г============T===============¬                              ');

writeln(m,'    ¦   Длина    ¦  Среднее время¦                              ');

writeln(m,'    ¦  очереди   ¦  обслуживания ¦                              ');

writeln(m,'    ¦============+===============¦                              ');

writeln(m,'    ¦   0        ¦     5,5       ¦                              ');

writeln(m,'    ¦============+===============¦                              ');

writeln(m,'    ¦ 1 или 2    ¦     5,0       ¦                              ');

writeln(m,'    ¦============+===============¦                              ');

	writeln(m,'    ¦ 3,4 или 5  ¦     4,5       ¦                              ');

writeln(m,'    ¦============+===============¦                              ');

writeln(m,'    ¦ 6 и более  ¦     4,0       ¦                              ');

writeln(m,'    L============¦===============-                              ');

writeln(m,'Задание:                                                        ');

writeln(m,'     Построить модель, которая давала бы информацию об очереди  ');

writeln(m,'     автомобилей и времени простоя склада.                      ');

writeln(m,'     Выполнение моделирования должно охватывать 5 отдельных     ');

writeln(m,'     пятичасовых дней.                                          ');

writeln(m,' ');

writeln(m,' Результаты испытания модели обслуживания автомобилей');

writeln(m,'                    в складе                   ');

writeln(m,'----------------T----------------------------------¬');

writeln(m,'¦Наименование   ¦          День недели             ¦');

writeln(m,'¦параметров     +------T------T------T------T------+');

writeln(m,'¦               ¦  Пн. ¦ Вт.  ¦  Ср. ¦  Чт. ¦  Пт. ¦');

writeln(m,'L---------------+------+------+------+------+-------');

{Задание начального значения среднего времени простоя складов}

      for i:=1 to Wday do Rday[i]:=0;

{Вывод времени простоя склада  по дням}

{writeln(m,' Время простоя,час');}

      for i:=1 to SkladCnt do

         begin

{           writeln(m,'  ',i,'-го пункта ');

            write(m,' обслужив. ');

            write(m,'склада');

}           for Id:=1 to Wday do

               begin

	{                write(m,SkladN[i,Id+1]/3600:6:2,' ');

}{Подсчёт среднего времени простоя складов по дням}

                Rday[Id]:=Rday[Id]+SkladN[i,Id+1];

               end;

{            writeln(m,'');

}         end;

{writeln(m,'----------------------------------------------------');

}{Вывод динамики очереди автомобилей по дням}

writeln(m,' Динамика очереди автомобилей');

for i:=1 to PntNum  do

         begin

            write(m,' через ',i*PntNumSec/3600:4:2,' часа ');

            for Id:=1 to Wday do write(m,AvtoMaxPnt[Id,i]:6,' ');

            writeln(m,'');

         end;

writeln(m,'----------------------------------------------------');

{Вывод количества автомобилей, приезжавших склад по дням}

write(m,' Количество      ');

      for Id:=1 to Wday do write(m,Avto[Id]:6,' ');

      writeln(m,'');

writeln(m,' автомобилей');

writeln(m,'----------------------------------------------------');

{Вывод максимального количества автомобилей в очереди  по дням}

write(m,' Максимальная    ');

      for Id:=1 to Wday do write(m,AvtoMax[Id]:6,' ');

      writeln(m,'');

writeln(m,' очередь ');

writeln(m,'----------------------------------------------------');

{Вывод среднего времени простоя склада по дням}

write(m,' Среднее время   ');

      for Id:=1 to Wday do write(m,Rday[Id]/(3600*SkladCnt):6:2,' ');

      writeln(m,'');

writeln(m,' простоя склада, час');

writeln(m,'----------------------------------------------------');

{Закрытие файла}

	close(m);

{Очистка экрана}

    clrScr;

{Вывод сообщения о сохранении файла}

     gotoxy(20,10);

     write('          ВАШ ФАЙЛ СОХРАНЕН !  ');

delay(20000);

   end;

{Раздел операторов}

begin

{Запуск процедуры моделирования процесса работы склада}

   model;

{Запуск процедуры вывода результатов работы программы}

   wywod

end.

4.3. Характеристика программы

Текст программы состоит из заголовка, раздела описаний, который содержит описание констант, переменных, функции датчика случайных чисел Rand, процедур:

· model, которая моделирует процесс работы склада,

· wywod, которая обеспечивает вывод результатов исследования модели в файл вывода, имя которого задается пользователем с клавиатуры;

и раздела операторов, состоящего из последовательности вызовов процедур model  и wywod.

Объём исходного текста программы вместе с комментариями составляет 314 строк

В результате компиляции программы получен исполняемый модуль объёмом 24896 байт (режим с сопроцессором). Присвоено имя исполняемой программе — Var14.exe. Программа не требует ввода исходных данных. Время исполнения программы на компьютере с процессором Pentium 200 не превышает 0,3 сек. Файл вывода результатов имеет объём не более 4 Кб. 

	4.4. Инструкция пользователю
Программа работает в режиме DOS. Первоначально записывают программу на компьютер в рабочую директорию или корневой каталог. Возможен запуск программы со сменных дисков, в том числе и  гибких.

Для запуска программы в командной строке набирают имя программы и нажимают клавишу <Enter>.

Пример

Записываем программу с именем Var14.exe в корневой каталог диска С. После чего  набирают в командной строке имя программы Var14 и нажимают клавишу <Enter> 

C:\>Var14<Enter>

В указанном примере C:\> приглашение DOS для ввода команды.

После запуска программы на экране появляется приглашение ввести имя файла для вывода результатов исследования в следующем виде.

                           ЗАПИСЬ НА ДИСК

         +---------------------------------------------------------+

         ¦ВВЕДИТЕ ИМЯ ФАЙЛА:                                 ¦

         +---------------------------------------------------------+

Введем имя файла privet
                           ЗАПИСЬ НА ДИСК

         +---------------------------------------------------------+

         ¦ВВЕДИТЕ ИМЯ ФАЙЛА:  privet                      ¦

         +---------------------------------------------------------+

и нажмём клавишу <Enter>.

На экране появится сообщение:

ВАШ ФАЙЛ СОХРАНЕН!

что является сообщением о корректном завершении программы.

Результаты работы программы в этом случае будут находиться в текстовом файле DOS  без расширения,  c именем privet. Полученный текст может быть просмотрен или распечатан с использованием имеющихся стандартных средств или редакторов поддерживающих работу с текстовыми файлами DOS.

{ copyright (с) Трушина М.А.,1999, МКУ и НТ, Москва

            Программа тестирования датчика случайных чисел

}

Program Test;

Uses Crt;

 Const

   WorkDayHour=5;                {Количество часов в рабочем дне}

	   AvtoDayMidl=200;            {Среднее количество автомобилей в день}

   WorkDaySec=WorkDayHour*60*60; {Количество секунд в рабочем дне}

   AvtoMidl=(60.0*60.0)/AvtoDayMidl;

                                 {Среднее время приезда  одного

                                 автомобиля на склад, в сек}

   SkladMidl=2.0*60.0;             {Среднее время обслуживания автомобиля

                                  на складе, в сек}

LambdaAvto=1.0/AvtoMidl;   {Среднее кол-во приехавших автомобилей на склад

                                  за одну сек}

Rw:=TimeSer*TimeSer*TimeSer;

	           Rc:=Rc+Rw;

           Rd:=Rd+Rw*TimeSer;

        end;

     M1:=Ra/I;M2:=Rb/I;M3:=Rc/i;M4:=Rd/i;

     D1:=M2-M1*M1;D2:=D1*i/(i-1);

     Writeln('Количество расчётов = ',i:10);

     Writeln('Среднее расчётное =',M1:10:5,

             ' ожидаемое =',1.0/LambdaSklad:10:5);

     Writeln('Дисперс.расчётная =',D2:10:5,

             ' ожидаемая =',1.0/(LambdaSklad*LambdaSklad):10:5);

    end.

	Результаты расчета имеют вид.

Количество расчётов =      18001

Среднее расчётное = 118.21199 ожидаемое = 120.00000

Дисперс.расчётная =14009.37793 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 119.84034 ожидаемое = 120.00000

Дисперс.расчётная =14359.58691 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 119.50261 ожидаемое = 120.00000

Дисперс.расчётная =14284.84766 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 120.79183 ожидаемое = 120.00000

Дисперс.расчётная =14423.79492 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 120.52113 ожидаемое = 120.00000

Дисперс.расчётная =14658.76953 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 118.07029 ожидаемое = 120.00000

Дисперс.расчётная =13702.14941 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 120.28457 ожидаемое = 120.00000

Дисперс.расчётная =14494.76953 ожидаемая =14400.00000

Количество расчётов =      18001

Среднее расчётное = 120.38284 ожидаемое = 120.00000

Дисперс.расчётная =14356.93652 ожидаемая =14400.00000

Анализ результатов расчёта показывает достаточную точность датчика случайных чисел, что позволило включить его в программу.

На втором этапе отладки программы, при исследовании её работоспособности, возникла ошибка «лишнего номера контрольной точки», которая была связана с округлением при расчёте интервала усреднения контрольной точки. Для её устранения в программу был вписаны операторы анализа и корректировки, которые устраняли выдачу лишнего номера контрольной точки на правой границе рабочего дня
После устранения указанной ошибки в период дальнейшего использования сбоев в её работе не наблюдалось.

	6. Полученные результаты

Один из результатов работы программы имеет следующий вид.

Курсовой проект по компьютерному моделированию      

Тема:                                                           

     Построение модели обслуживания автомобилей на складе       

Описание модели:                                                

     Автомобили с товаром подъезжают к складу с интенсивностью  

     12 автомобилей в час в соответствии с пуассоновским законом

     распределения.                                             

     Обслуживание является экспоненциальным, но среднее время   

     обслуживания зависит от числа автомобилей, находящихся в   

     очереди.                                                   

     Эта зависимость представлена в таблице:                    

    г============T===============¬                              

    ¦   Длина    ¦  Среднее время¦                              

    ¦  очереди   ¦  обслуживания ¦                              

    ¦============+===============¦                              

    ¦   0        ¦     5,5       ¦                              

    ¦============+===============¦                              

    ¦ 1 или 2    ¦     5,0       ¦                              

    ¦============+===============¦                              

    ¦ 3,4 или 5  ¦     4,5       ¦                              

    ¦============+===============¦                              

    ¦ 6 и более  ¦     4,0       ¦                              

    L============¦===============-                              

Задание:                                                        

     Построить модель, которая давала бы информацию об очереди  

     автомобилей и времени простоя склада.                      

     Выполнение моделирования должно охватывать 5 отдельных     

     пятичасовых дней.                                          

 Результаты испытания модели обслуживания автомобилей

                    в складе                   

----------------T----------------------------------¬

¦Наименование   ¦          День недели             ¦

¦параметров     +------T------T------T------T------+

¦               ¦  Пн. ¦ Вт.  ¦  Ср. ¦  Чт. ¦  Пт. ¦

L---------------+------+------+------+------+-------

Динамика очереди автомобилей

 через 0.25 часа      1      1      1      2      0 

 через 0.50 часа      1      1      1      2      1 

 через 0.75 часа      1      2      3      2      4 

 через 1.00 часа      1      2      5      1      5 

	 через 1.25 часа      1      6      1      3      5 

 через 1.50 часа      1      6      4      2      1 

 через 1.75 часа      1      3      7      1      3 

 через 2.00 часа      2      2     13      1      5 

 через 2.25 часа      6      4     13      2      2 

 через 2.50 часа      5      2     13      4      5 

 через 2.75 часа      4      1     10     11      6 

 через 3.00 часа      4      1     13     10      3 

 через 3.25 часа      5      1     12      5      1 

 через 3.50 часа      4      2     13      1      2 

 через 3.75 часа      4      4      7      2      3 

 через 4.00 часа      4      3      7      7      6 

 через 4.25 часа      3      1      7      7      5 

 через 4.50 часа      3      1      5      5      3 

 через 4.75 часа      5      2      5      1      3 

 через 5.00 часа      5      3      2      6      3 

----------------------------------------------------

 Количество          54     54     66     64     60 

 Автомобилей

----------------------------------------------------

 Максимальная         6      6     13     11      6 

 очередь 

----------------------------------------------------

 Среднее время     1.08   1.21   0.25   0.68   0.81 

 простоя склада, час

----------------------------------------------------

	7. Заключение

       В результат выполнения курсовой работы разработан алгоритм, соответствующий: программе c именем Sklad, программа функции случайных чисел Rand, программа процедуре моделирования деятельности склада model, программа процедуры вывода результатов  wywod. При написании программ был использован язык Turbo Pascal 5.0.

Результаты работы программы выводятся в файл, имя которого задаёт пользователь.

Результаты работы программы приведены в разделе 6, что показало её отлаженность, работоспособность и соответствие заданию на курсовой проект.

